Overview

- Boltzmann equation, conservation laws
- Hermite polynomial expansions
- Equilibrium distribution

- $f(\boldsymbol{x},\boldsymbol{\xi},t)$ probability distribution function
- Force-free Boltzmann equation (dimensionless)

$$\frac{\partial f}{\partial t}+\xi_\alpha\frac{\partial f}{\partial x_\alpha}=\Omega(f)$$

- Conservation laws

$$\begin{aligned}
&\text{Mass} && \int f(\boldsymbol{x},\boldsymbol{\xi},t)\,\mathrm{d}\boldsymbol{\xi}= && \int f^{\mathrm{eq}}(\boldsymbol{x},\boldsymbol{\xi},t)\,\mathrm{d}\boldsymbol{\xi}= && \rho(\boldsymbol{x},t)\\
&\text{Momentum} && \int f(\boldsymbol{x},\boldsymbol{\xi},t)\boldsymbol{\xi}\,\mathrm{d}\boldsymbol{\xi}= && \int f^{\mathrm{eq}}(\boldsymbol{x},\boldsymbol{\xi},t)\boldsymbol{\xi}\,\mathrm{d}\boldsymbol{\xi}= && \rho\boldsymbol{u}(\boldsymbol{x},t)\\
&\text{Total energy} && \frac{1}{2}\int f(\boldsymbol{x},\boldsymbol{\xi},t)|\boldsymbol{\xi}|^2\,\mathrm{d}\boldsymbol{\xi}= && \frac{1}{2}\int f^{\mathrm{eq}}(\boldsymbol{x},\boldsymbol{\xi},t)|\boldsymbol{\xi}|^2\,\mathrm{d}\boldsymbol{\xi}= && \rho E(\boldsymbol{x},t)\\
&\text{Internal energy} && \frac{1}{2}\int f(\boldsymbol{x},\boldsymbol{\xi},t)|\boldsymbol{\xi}-\boldsymbol{u}|^2\,\mathrm{d}\boldsymbol{\xi}= && \frac{1}{2}\int f^{\mathrm{eq}}(\boldsymbol{x},\boldsymbol{\xi},t)|\boldsymbol{\xi}-\boldsymbol{u}|^2\,\mathrm{d}\boldsymbol{\xi}= && \rho e(\boldsymbol{x},t)
\end{aligned}$$

$$E=e+\frac{1}{2}|\boldsymbol{u}|^2$$

- Definition from generating (weight) function

$$\omega(x) = \frac{1}{\sqrt{2\pi}} \exp\left(-\frac{x^2}{2}\right), H^{(n)}(x) = (-1)^n \frac{1}{\omega(x)} \frac{\mathrm{d}^n}{\mathrm{d}x^n} \omega(x)$$

```
In[1]:= omega[x_] = Exp[-x^2/2]/Sqrt[2Pi]
```

$$\frac{e^{-\frac{x^2}{2}}}{\sqrt{2\,\pi}}$$

```
In[2]:= H[n_,x_]:=Expand[(-1)^n/omega[x] D[omega[x],{x,n}]];
        Table[H[n,x],{n,0,5}]
```

$$\{1, x, x^2 - 1, x^3 - 3\,x, x^4 - 6\,x^2 + 3, x^5 - 10\,x^3 + 15\,x\}$$

```
In[3]:= HHd1=Table[Integrate[omega[x] H[i,x] H[j,x],{x,-Infinity,Infinity}],
                  {i,0,4},{j,0,4}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 2 & 0 & 0 \\ 0 & 0 & 0 & 6 & 0 \\ 0 & 0 & 0 & 0 & 24 \end{pmatrix}$$

```
In[4]:=
```

- Define multivariate generating function

$$\omega(\boldsymbol{x}) = \frac{1}{(2\pi)^{d/2}} \exp\left(-\frac{\boldsymbol{x}^2}{2}\right), \boldsymbol{H}^{(n)}(\boldsymbol{x}) = (-1)^n \frac{1}{\omega(\boldsymbol{x})} \nabla^{(n)} \omega(\boldsymbol{x}), \nabla^{(n)} = \nabla \otimes \ldots \otimes \nabla$$

$$\nabla^{(1)} = \nabla = \begin{pmatrix} \frac{\partial}{\partial x_1} \\ \vdots \\ \frac{\partial}{\partial x_d} \end{pmatrix}, \nabla^{(2)} = \nabla \otimes \nabla = \begin{pmatrix} \frac{\partial^2}{\partial x_1} & \cdots & \frac{\partial^2}{\partial x_1 \partial x_d} \\ \vdots & \ddots & \vdots \\ \frac{\partial^2}{\partial x_d \partial x_1} & \cdots & \frac{\partial^2}{\partial x_d \partial x_d} \end{pmatrix}$$

```
In[4]:= omega[x_,y_,z_] =  Exp[-(x^2+y^2+z^2)/2]/(2Pi)^(3/2)
```

$$\frac{e^{\frac{1}{2}(-x^2-y^2-z^2)}}{2\sqrt{2}\,\pi^{3/2}}$$

```
In[5]:=
```

```
In[5]:= H[n_,i_,j_,k_] :=
          Simplify[(-1)^n/omega[x,y,z] D[omega[x,y,z],{x,i},{y,j},{z,k}]];
        H[n_]:= H[n] = DeleteCases[Flatten[Table[If[i+j+k==n,H[n,i,j,k],{}],
                                 {k,0,n},{j,0,n},{i,0,n}]],{}];
        H[0]
```

$\{1\}$

```
In[6]:= H[1]
```

$\{x, y, z\}$

```
In[7]:= H[2]
```

$\{x^2 - 1, x\, y, y^2 - 1, x\, z, y\, z, z^2 - 1\}$

```
In[8]:= H[3]
```

$\{x\,(x^2 - 3), (x^2 - 1)\, y, x\,(y^2 - 1), y\,(y^2 - 3), (x^2 - 1)\, z, x\, y\, z, (y^2 - 1)\, z, x\,(z^2 - 1), y\,(z^2 - 1), z\,(z^2 - 3)\}$

```
In[9]:=
```

$$\int \omega(\boldsymbol{x})\, H_{\boldsymbol{\alpha}}^{(m)}(\boldsymbol{x})\, H_{\boldsymbol{\beta}}^{(n)}(\boldsymbol{x})\, \mathrm{d}\boldsymbol{x} = n_x! n_y! n_z! \delta_{mn}^{(2)} \delta_{\boldsymbol{\alpha\beta}}^{(m+n)},\ \delta_{\boldsymbol{\alpha\beta}}^{(m+n)} = 1 \text{ if } \boldsymbol{\alpha} \text{ is permutation of } \boldsymbol{\beta}$$

$$\boldsymbol{\alpha} = (x, x, y) \Rightarrow (n_x, n_y, n_z) = (2, 1, 0).$$

Hermite expansion of equilibrium distribution

- Seek Hermite expansion of equilibrium function ($d=3$)

$$f^{\mathrm{eq}}(\boldsymbol{\xi};\rho,\boldsymbol{u},\theta)=\frac{\rho}{(2\pi\theta)^{d/2}}\exp\left[-\frac{(\boldsymbol{\xi}-\boldsymbol{u})^2}{2\theta}\right]=\omega(\boldsymbol{\xi})\sum_{n=0}^{\infty}\frac{1}{n!}\boldsymbol{a}^{(n)}(\rho,\boldsymbol{u},\theta)\cdot\boldsymbol{H}^{(n)}(\boldsymbol{\xi})$$

$$\boldsymbol{a}^{(n)}(\rho,\boldsymbol{u},\theta)=\int f^{\mathrm{eq}}(\boldsymbol{\xi};\rho,\boldsymbol{u},\theta)\,\boldsymbol{H}^{(n)}(\boldsymbol{\xi})\,\mathrm{d}\boldsymbol{\xi}$$

- $f^{\mathrm{eq}}(\boldsymbol{\xi})$ of same form as $\omega(\boldsymbol{\xi})=\exp(-\boldsymbol{x}^2/2)/(2\pi)^{d/2}$

$$f^{\mathrm{eq}}(\boldsymbol{\xi};\rho,\boldsymbol{u},\theta)=\frac{\rho}{(2\pi\theta)^{d/2}}\exp\left[-\frac{(\boldsymbol{\xi}-\boldsymbol{u})^2}{2\theta}\right]=\frac{\rho}{\theta^{d/2}}\omega\left(\frac{\boldsymbol{\xi}-\boldsymbol{u}}{\sqrt{\theta}}\right)$$

- Compute coefficients $\boldsymbol{a}^{(n)}(\rho,\boldsymbol{u},\theta)$, $\boldsymbol{\eta}=(\boldsymbol{\xi}-\boldsymbol{u})/\sqrt{\theta}$ (scale to thermal velocity)

$$\boldsymbol{a}^{(n),\mathrm{eq}}(\rho,\boldsymbol{u},\theta)=\frac{\rho}{\theta^{d/2}}\int\omega\left(\frac{\boldsymbol{\xi}-\boldsymbol{u}}{\sqrt{\theta}}\right)\boldsymbol{H}^{(n)}(\boldsymbol{\xi})\,\mathrm{d}\boldsymbol{\xi}=\rho\int\omega(\boldsymbol{\eta})\boldsymbol{H}^{(n)}(\sqrt{\theta}\,\boldsymbol{\eta}+\boldsymbol{u})\,\mathrm{d}\boldsymbol{\eta}$$

$$\boldsymbol{a}^{(n),\mathrm{eq}}(\rho,\boldsymbol{u},\theta) = \frac{\rho}{\theta^{d/2}} \int \omega\left(\frac{\boldsymbol{\xi}-\boldsymbol{u}}{\sqrt{\theta}}\right) \boldsymbol{H}^{(n)}(\boldsymbol{\xi})\,\mathrm{d}\boldsymbol{\xi} = \rho \int \omega(\boldsymbol{\eta}) \boldsymbol{H}^{(n)}(\sqrt{\theta}\,\boldsymbol{\eta}+\boldsymbol{u})\,\mathrm{d}\boldsymbol{\eta}$$

```
In[9]:= aeq[n_] := aeq[n] = Module[{Hn,an,nH,HnE,inf,st},
          Hn=H[n]; nH=Length[Hn]; inf=Infinity; st=Sqrt[theta];
          HnE = Hn /. {x->st r + u, y->st s + v, z->st t + w};
          an=Table[Integrate[ omega[r,s,t] HnE[[k]],
                   {r,-inf,inf},{s,-inf,inf},{t,-inf,inf}],{k,1,nH}];
          Return[rho an] ];
        Aeq={aeq[0],aeq[1]}
```

$\{\{\rho\},\{\rho\,u,\rho\,v,\rho\,w\}\}$

```
In[10]:= AppendTo[Aeq,aeq[2]]
```

$\{\{\rho\},\{\rho\,u,\rho\,v,\rho\,w\},\{\rho\,(\theta+u^2-1),\rho\,u\,v,\rho\,(\theta+v^2-1),\rho\,u\,w,\rho\,v\,w,\rho\,(\theta+w^2-1)\}\}$

```
In[11]:= Simplify[Aeq[[3]][[1]]+Aeq[[3]][[3]]+Aeq[[3]][[6]]]
```

$\rho\,(3\,\theta+u^2+v^2+w^2-3)$

```
In[12]:=
```

```
In[12]:= AppendTo[Aeq,aeq[3]];
         Aeq[[1]]
```

$\{\rho\}$

```
In[13]:= Aeq[[2]]
```

$\{\rho\, u, \rho\, v, \rho\, w\}$

```
In[14]:= Aeq[[3]]
```

$\{\rho\,(\theta + u^2 - 1), \rho\, u\, v, \rho\,(\theta + v^2 - 1), \rho\, u\, w, \rho\, v\, w, \rho\,(\theta + w^2 - 1)\}$

```
In[15]:= Aeq[[4]]
```

$\{\rho\, u\,(3\,\theta + u^2 - 3), \rho\, v\,(\theta + u^2 - 1), \rho\, u\,(\theta + v^2 - 1), \rho\, v\,(3\,\theta + v^2 - 3), \rho\, w\,(\theta + u^2 - 1), \rho\, u\, v\, w, \rho\, w\,(\theta + v^2 - 1), \rho\, u\,(\theta + w^2 - 1), \rho\, v\,(\theta + w^2 - 1), \rho\, w\,(3\,\theta + w^2 - 3)\}$

```
In[16]:=
```

First three terms of Hermite expansion correspond to conserved quantities

```
In[16]:= vRules = {x->xi, y->eta, z->zeta};
         feqH[xi_,eta_,zeta_] = omega[xi,eta,zeta] Sum[aeq[n].H[n]/n! /. vRules, {n,0,2}]
```

$$e^{\frac{1}{2}(-\eta^2-\xi^2-\zeta^2)}\left(\frac{1}{2}\left(\left(\eta^2-1\right)\rho\left(\theta+v^2-1\right)+\eta\,\rho\,u\,v\,\xi+\eta\,\rho\,v\,w\,\zeta+\rho\left(\xi^2-1\right)\left(\theta+u^2-1\right)+\rho\left(\zeta^2-1\right)\left(\theta+w^2-1\right)+\rho\,u\,w\,\xi\,\zeta\right)+\eta\,\rho\,v+\rho\,u\,\xi+\rho\,w\,\zeta+\rho\right)\Big/\left(2\sqrt{2}\,\pi^{3/2}\right)$$

```
In[18]:= inf=Infinity;
         Integrate[feqH[xi,eta,zeta],{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,inf}]
```

ρ

```
In[19]:= Integrate[feqH[xi,eta,zeta] xi,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,inf}]
```

$\rho\,u$

```
In[20]:= Integrate[feqH[xi,eta,zeta] eta,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,inf}]
```

$\rho\,v$

```
In[21]:= Integrate[feqH[xi,eta,zeta] zeta,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,inf}]
```

$\rho\,w$

```
In[22]:=
```

Hermite expansion of equilibrium distribution function, verify 2^{nd} moments, diagonal

```
In[23]:= Integrate[feqH[xi,eta,zeta] xi xi,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,inf}]
```

$$\rho\left(\theta + u^2\right)$$

```
In[24]:= Integrate[feqH[xi,eta,zeta] eta eta,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,
         inf}]
```

$$\rho\left(\theta + v^2\right)$$

```
In[25]:= Integrate[feqH[xi,eta,zeta] zeta zeta,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,
         inf}]
```

$$\rho\left(\theta + w^2\right)$$

Hermite expansion of equilibrium distribution function, verify 2^{nd} moments, off-diagonal

```
In[26]:= Integrate[feqH[xi,eta,zeta] xi eta,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,inf}]
```

$$\frac{\rho\, u\, v}{2}$$

```
In[27]:= Integrate[feqH[xi,eta,zeta] xi zeta,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,
         inf}]
```

$$\frac{\rho\, u\, w}{2}$$

```
In[28]:= Integrate[feqH[xi,eta,zeta] eta zeta,{xi,-inf,inf},{eta,-inf,inf},{zeta,-inf,
         inf}]
```

$$\frac{\rho\, v\, w}{2}$$

- Verify $\boldsymbol{a}^{(n),\mathrm{eq}} = \boldsymbol{a}^{(n),\mathrm{eq}} \otimes \boldsymbol{u}$, $a^{(0),\mathrm{eq}} = \rho$

```
In[29]:= V={u,v,w}; a0[1] = aeq[1] /. rho -> Subscript[rho,0]
```

$\{\rho_0\, u, \rho_0\, v, \rho_0\, w\}$

```
In[30]:= a0[2] = Outer[Times,a0[1],V]
```

$$\begin{pmatrix} \rho_0\, u^2 & \rho_0\, u\, v & \rho_0\, u\, w \\ \rho_0\, u\, v & \rho_0\, v^2 & \rho_0\, v\, w \\ \rho_0\, u\, w & \rho_0\, v\, w & \rho_0\, w^2 \end{pmatrix}$$

```
In[31]:= aeq[2] /. theta->1
```

$\{\rho\, u^2, \rho\, u\, v, \rho\, v^2, \rho\, u\, w, \rho\, v\, w, \rho\, w^2\}$

```
In[32]:=
```

- Third-order

```
In[32]:= a0[3] = DeleteDuplicates[Flatten[Outer[Times,a0[2],V]]]
```

$\{\rho_0\,u^3, \rho_0\,u^2\,v, \rho_0\,u^2\,w, \rho_0\,u\,v^2, \rho_0\,u\,v\,w, \rho_0\,u\,w^2, \rho_0\,v^3, \rho_0\,v^2\,w, \rho_0\,v\,w^2, \rho_0\,w^3\}$

```
In[33]:= aeq[3] /. theta->1
```

$\{\rho\,u^3, \rho\,u^2\,v, \rho\,u\,v^2, \rho\,v^3, \rho\,u^2\,w, \rho\,u\,v\,w, \rho\,v^2\,w, \rho\,u\,w^2, \rho\,v\,w^2, \rho\,w^3\}$

```
In[34]:=
```

- $\boldsymbol{a}^{(n),\mathrm{eq}}(\rho,\boldsymbol{u},\theta) = \int f^{\mathrm{eq}}(\boldsymbol{\xi})\boldsymbol{H}^{(n)}(\boldsymbol{\xi})\,\mathrm{d}\boldsymbol{\xi},\; f^{\mathrm{eq},N}(\boldsymbol{\xi}) = \omega(\boldsymbol{\xi})\,\rho Q(\boldsymbol{u},\theta,\boldsymbol{\xi}) \Rightarrow$

$$\boldsymbol{a}^{(n),\mathrm{eq}}(\rho,\boldsymbol{u},\theta) = \rho\int\omega(\boldsymbol{\xi})\,Q(\boldsymbol{u},\theta,\boldsymbol{\xi})\,\boldsymbol{H}^{(n)}(\boldsymbol{\xi})\,\mathrm{d}\boldsymbol{\xi} = \sum_{i=0}^{[(n+2)/2]} w_i\,Q(\boldsymbol{u},\theta,\boldsymbol{\xi}_i)\,\boldsymbol{H}^{(n)}(\boldsymbol{\xi}_i)$$

```
In[34]:= H[2,x]
```

$x^2 - 1$

```
In[35]:=
```

$$\int_{-\infty}^{\infty} w(x)\,f(x)\,\mathrm{d}x \cong \sum_{i=0}^{q-1} w_i\,f(x_i),\;\int_{-\infty}^{\infty} w(x)\,p_{2n-1}(x)\,\mathrm{d}x = \sum_{i=0}^{n-1} w_i\,p_n(x_i)$$

- Consider a discrete velocity set $\{\boldsymbol{\xi}_0, \ldots, \boldsymbol{\xi}_q\}$. The Hermite expansion of the Maxwell-Boltzmann distribution up to second order $N=2$ is

$$f^{\mathrm{eq}}(\boldsymbol{\xi};\rho,\boldsymbol{u},\theta) \cong f^{\mathrm{eq},N}(\boldsymbol{\xi};\rho,\boldsymbol{u},\theta) = \omega(\boldsymbol{\xi})\,\rho\,Q(\boldsymbol{\xi};\boldsymbol{u},\theta)$$

- Evaluate the $N=2$ approximation for discrete velocity $\boldsymbol{\xi}_i$

$$f_i^{\mathrm{eq}} = \rho\,w_i\,Q(\boldsymbol{\xi}_i;\boldsymbol{u},\theta),\, w_i \equiv \omega(\boldsymbol{\xi}_i)$$

```
In[39]:= Q[xi_,eta_,zeta_]=Simplify[feqH[xi,eta,zeta]/omega[xi,eta,zeta]/rho]
```

$$\frac{1}{2}\left(\eta^2\left(\theta+v^2-1\right)+\eta\,v\left(u\,\xi+w\,\zeta+2\right)+\theta\left(\xi^2+\zeta^2-3\right)+u^2\,\xi^2-u^2+u\,w\,\xi\,\zeta+2\,u\,\xi-v^2+w^2\,\zeta^2-w^2+2\,w\,\zeta-\xi^2-\zeta^2+5\right)$$

```
In[47]:= Q[xi,0,0] /. theta->1
```

$$\frac{1}{2}\left(u^2\,\xi^2-u^2+2\,u\,\xi-v^2-w^2+2\right)$$

```
In[48]:=
```

- For a discrete set of velocities $\{\boldsymbol{c}_i\}_{i=0,1,\ldots,Q}$ the discrete Boltzmann equation is

$$\partial_t f_i + c_{i\alpha} \partial_\alpha f_i = \Omega(f_i), i = 0, 1, \ldots, Q,$$

$$\partial_t f_i + (\boldsymbol{c}_i \cdot \nabla) f_i = \Omega(f_i)$$

- Step 1 of operator splitting, $\partial_t f_i + (\boldsymbol{c}_i \cdot \nabla) f_i = 0$,

$$f_i(t + \delta t, \boldsymbol{x}) = f_i(t, \boldsymbol{x} - \boldsymbol{c}_i \delta t)$$

- Step 2 of operator splitting

$$\partial_t f_i = \Omega(f_i)$$

$$\frac{\mathrm{d}f_i}{\mathrm{d}t} = -\frac{\left(f_i - f_i^{(0)}\right)}{\tau} \Rightarrow f_i(1) = e^{-1/\tau} f_i(0) + f_i^{\mathrm{eq}}(1 - e^{-1/\tau}) = e^{-1/\tau}(f_i(0) - f_i^{\mathrm{eq}}) + f_i^{\mathrm{eq}}$$

$$f_i(1) = \left(1 - \frac{1}{\tau} + \frac{1}{2!\tau^2} - \frac{1}{3!\tau^3} + \cdots\right)(f_i(0) - f_i^{\mathrm{eq}}) + f_i^{\mathrm{eq}} \cong \left(1 - \frac{1}{\tau}\right)(f_i(0) - f_i^{\mathrm{eq}}) + f_i^{\mathrm{eq}}$$

$$f_i(1) = f_i(0) + \frac{1}{\tau}(f_i^{\mathrm{eq}} - f_i(0))$$

```
In[50]:= Simplify[DSolve[{fi'[t]==-(fi[t]-fieq)/tau,fi[0]==fi0},fi[t],t]]
```

$$\left\{\left\{\mathrm{fi}(t) \to e^{-\frac{t}{\tau}}(\mathrm{fi0} + \mathrm{fieq}(e^{t/\tau} - 1))\right\}\right\}$$

```
In[51]:=
```

$$y' = f(y) = \frac{\partial f}{\partial y} y, |1 + z| \leqslant 1, z = \lambda \delta t, y \sim e^{\lambda t}$$